Ricerca Operativa - Laboratorio

Lezione 4 - Dualità in PL con AMPL

Docente: Luigi De Giovanni

Dipartimento di Matematica "Tullio Levi-Civita"
Università degli Studi di Padova

luigi@math.unipd.it
https://www.math.unipd.it/∼luigi/

Corso di Laurea Magistrale in Matematica
Università degli Studi di Padova
a.a. 2019–2020

Teoria della dualità nella PL (i)

Dati:

$c \in \mathbb{R}^p$, $d \in \mathbb{R}^k$, $C \in \mathbb{R}^{q \times p}$, $D \in \mathbb{R}^{q \times k}$, $h \in \mathbb{R}^q$, $E \in \mathbb{R}^{m \times p}$, $F \in \mathbb{R}^{m \times k}$, $g \in \mathbb{R}^m$

PROBLEMA PRIMALE	**PROBLEMA DUALE**
$\min\limits_{\substack{x \in \mathbb{R}^p \\ y \in \mathbb{R}^k}} \; c^T x + d^T y$ $Cx + Dy = h$ $Ex + Fy \geq g$ $x \geq 0$	

Teoria della dualità nella PL (i)

Dati:
$c \in \mathbb{R}^p$, $d \in \mathbb{R}^k$, $C \in \mathbb{R}^{q \times p}$, $D \in \mathbb{R}^{q \times k}$, $h \in \mathbb{R}^q$, $E \in \mathbb{R}^{m \times p}$, $F \in \mathbb{R}^{m \times k}$, $g \in \mathbb{R}^m$

PROBLEMA PRIMALE	PROBLEMA DUALE
$\min\limits_{\substack{x \in \mathbb{R}^p \\ y \in \mathbb{R}^k}} \; c^T x + d^T y$	$\max\limits_{\substack{u \in \mathbb{R}^q \\ v \in \mathbb{R}^m}} \; h^T u + g^T v$
$Cx + Dy = h$	$C^T u + E^T v \leq c$
$Ex + Fy \geq g$	$D^T u + F^T v = d$
$x \geq 0$	$v \geq 0$

Teoria della dualità nella PL (i)

Dati:
$c \in \mathbb{R}^p$, $d \in \mathbb{R}^k$, $C \in \mathbb{R}^{q \times p}$, $D \in \mathbb{R}^{q \times k}$, $h \in \mathbb{R}^q$, $E \in \mathbb{R}^{m \times p}$, $F \in \mathbb{R}^{m \times k}$, $g \in \mathbb{R}^m$

PROBLEMA PRIMALE	**PROBLEMA DUALE**
$\min\limits_{\substack{x \in \mathbb{R}^p \\ y \in \mathbb{R}^k}} \; c^T x + d^T y$	$\max\limits_{\substack{u \in \mathbb{R}^q \\ v \in \mathbb{R}^m}} \; h^T u + g^T v$
$Cx + Dy = h$	$C^T u + E^T v \leq c$
$Ex + Fy \geq g$	$D^T u + F^T v = d$
$x \geq 0$	$v \geq 0$

Il duale del Problema Duale è il Problema Primale.

Teoria della dualità nella PL (ii)

Dualità debole

Siano $(\bar{x}, \bar{y})$ un punto ammissibile per il Problema Primale e $(\bar{u}, \bar{v})$ un punto ammissibile per il Problema Duale. Allora,

$$c^T \bar{x} + d^T \bar{y} \geq h^T \bar{u} + g^T \bar{v}.$$

Teoria della dualità nella PL (ii)

Dualità debole

Siano $(\bar{x}, \bar{y})$ un punto ammissibile per il Problema Primale e $(\bar{u}, \bar{v})$ un punto ammissibile per il Problema Duale. Allora,

$$c^T\bar{x} + d^T\bar{y} \geq h^T\bar{u} + g^T\bar{v}.$$

Dualità forte

- Il Problema Primale ammette soluzione ottima se e solo se il Problema Duale ammette soluzione ottima.
- Siano $(\bar{x}, \bar{y})$ un punto ammissibile per il Problema Primale e $(\bar{u}, \bar{v})$ un punto ammissibile per il Problema Duale. I due punti sono soluzioni ottime dei rispettivi problemi se e solo se

$$c^T\bar{x} + d^T\bar{y} = h^T\bar{u} + g^T\bar{v}.$$

Dualità e analisi di sensitività (i)

Consideriamo un generico problema di PL in forma standard:

$$\begin{aligned} \max\ & c^T x \\ Ax &= b \\ x &\geq 0 \end{aligned} \tag{P1}$$

Siano $x^* = \begin{bmatrix} A_B^{-1} b \\ 0 \end{bmatrix}$ la soluzione ottima di (P1) e $u^* = (A_B^{-1})^T c_B$ la soluzione ottima del duale di (P1), dove B è una base di A.

Dualità e analisi di sensitività (i)

Consideriamo un generico problema di PL in forma standard:

$$\begin{aligned} \max\ & c^T x \\ Ax &= b \\ x &\geq 0 \end{aligned} \tag{P1}$$

Siano $x^* = \begin{bmatrix} A_B^{-1} b \\ 0 \end{bmatrix}$ la soluzione ottima di (P1) e $u^* = (A_B^{-1})^T c_B$ la soluzione ottima del duale di (P1), dove B è una base di A.

Perturbando b_i di una quantità ϵ (per un certo indice $i \in \{1, \dots, n\}$), **se B è ancora una base ottima per il problema perturbato**, otteniamo

$$c^T \bar{x} - c^T x^* = \epsilon u_i^*,$$

dove $\bar{x}$ è la soluzione ottima del problema perturbato (mentre la soluzione ottima del nuovo problema duale rimane u^*).

Dualità e analisi di sensitività (i)

Consideriamo un generico problema di PL in forma standard:

$$\begin{aligned} \max\ & c^T x \\ & Ax = b \\ & x \geq 0 \end{aligned} \tag{P1}$$

Siano $x^* = \begin{bmatrix} A_B^{-1} b \\ 0 \end{bmatrix}$ la soluzione ottima di (P1) e $u^* = (A_B^{-1})^T c_B$ la soluzione ottima del duale di (P1), dove B è una base di A.

Perturbando b_i di una quantità ϵ (per un certo indice $i \in \{1, \ldots, n\}$), **se B è ancora una base ottima per il problema perturbato**, otteniamo

$$c^T \bar{x} - c^T x^* = \epsilon u_i^*,$$

dove $\bar{x}$ è la soluzione ottima del problema perturbato (mentre la soluzione ottima del nuovo problema duale rimane u^*).

La base B resta ottima se ϵ cade dentro un certo intervallo contente lo 0, ottenuto risolvendo

$$\epsilon A_B^{-1} e_i \geq -\bar{x}_B$$

Dualità e analisi di sensitività (ii)

Le variabili duali u_i^* possono essere interpretate come prezzi associati a un incremento unitario dei termini noti b_i.

Dualità e analisi di sensitività (ii)

Le variabili duali u_i^* possono essere interpretate come prezzi associati a un incremento unitario dei termini noti b_i.

Qual è il costo massimo unitario che sono disposto a sostenere per acquistare ulteriori quantità della risorsa i-esima? [nell'ipotesi che la base ottima non cambi...]

Dualità e analisi di sensitività (ii)

Le variabili duali u_i^* possono essere interpretate come prezzi associati a un incremento unitario dei termini noti b_i.

Qual è il costo massimo unitario che sono disposto a sostenere per acquistare ulteriori quantità della risorsa i-esima? [nell'ipotesi che la base ottima non cambi...]

Al più u_i^*! Perchè è l'incremento unitario che otterrei nella funzione obiettivo.

Dualità e analisi di sensitività (iii)

Esempio

Un'industria produce due tipi di prodotti: *de luxe* e *standard*. Per avere un prodotto finito di ciascuno dei due tipi sono necessari due ingredienti grezzi I_1 e I_2 e la lavorazione su una macchina. La tabella che segue riporta le quantità in Kg di ciascuno degli ingredienti e le ore di lavorazione sulla macchina necessarie per ottenere un prodotto finito di ciascuno dei due tipi.

	De luxe	Standard
I_1	3	2
I_2	4	1
ore lavorazione	2	1

Settimanalmente si hanno a disposizione al più 1200 Kg dell'ingrediente I_1 e al più 1000 Kg dell'ingrediente I_2, mentre la disponibilità massima settimanale di ore lavorative della macchina è pari a 700. Un prodotto de luxe è venduto a 24 Euro e un prodotto standard è venduto a 14 Euro. Si vuole pianificare la produzione settimanale in modo da massimizzare il profitto complessivo assumendo che i prodotti siano frazionabili.

Dualità e analisi di sensitività (iv)

PROBLEMA PRIMALE	PROBLEMA DUALE
$\max\ 24x_1 + 14x_2$	
$3x_1 + 2x_2 \leq 1200$	
$4x_1 + x_2 \leq 1000$	
$2x_1 + x_2 \leq 700$	
$x_1 \geq 0, x_2 \geq 0$	

Dualità e analisi di sensitività (iv)

PROBLEMA PRIMALE	PROBLEMA DUALE
$\max\ 24x_1 + 14x_2$	
$3x_1 + 2x_2 \leq 1200$	
$4x_1 + x_2 \leq 1000$	
$2x_1 + x_2 \leq 700$	
$x_1 \geq 0, x_2 \geq 0$	

E' conveniente per l'industria acquistare maggiori quantità di ingredienti I_1 e/o I_2? Se sì, a quale prezzo?

E' conveniente per l'industria aumentare il numero massimo di ore di lavorazione della macchina?

Dualità e analisi di sensitività (iv)

PROBLEMA PRIMALE	PROBLEMA DUALE
$\max\ 24x_1 + 14x_2$ $3x_1 + 2x_2 \leq 1200$ $4x_1 + x_2 \leq 1000$ $2x_1 + x_2 \leq 700$ $x_1 \geq 0, x_2 \geq 0$	$\min\ 1200u_1 + 1000u_2 + 700u_3$ $3u_1 + 4u_2 + 2u_3 \geq 24$ $2u_1 + u_2 + u_3 \geq 14$ $u_1 \geq 0, u_2 \geq 0, u_3 \geq 0$
$x_1^* = 160, \quad x_2^* = 360$	$u_1^* = 6.4, \quad u_2^* = 1.2, \quad u_3^* = 0$

E' conveniente per l'industria acquistare maggiori quantità di ingredienti I_1 e/o I_2? Se sì, a quale prezzo?

E' conveniente per l'industria aumentare il numero massimo di ore di lavorazione della macchina?

Dualità e analisi di sensitività (v)

- aumento di 1 kg della disponibilità di $I_1 \Rightarrow$ incremento di 6.4 € nel profitto complessivo

Dualità e analisi di sensitività (v)

- aumento di 1 kg della disponibilità di I_1 $\Rightarrow$ incremento di 6.4 € nel profitto complessivo
- aumento di 1 kg della disponibilità di I_2 $\Rightarrow$ incremento di 1.2 € nel profitto complessivo

Dualità e analisi di sensitività (v)

- aumento di 1 kg della disponibilità di I_1 $\Rightarrow$ incremento di 6.4 € nel profitto complessivo
- aumento di 1 kg della disponibilità di I_2 $\Rightarrow$ incremento di 1.2 € nel profitto complessivo
- aumento di 1 ora lavorativa $\Rightarrow$ incremento di 0 € nel profitto complessivo

Dualità e analisi di sensitività (v)

- aumento di 1 kg della disponibilità di I_1 $\Rightarrow$ incremento di 6.4 € nel profitto complessivo
- aumento di 1 kg della disponibilità di I_2 $\Rightarrow$ incremento di 1.2 € nel profitto complessivo
- aumento di 1 ora lavorativa $\Rightarrow$ incremento di 0 € nel profitto complessivo

Il prezzo ombra rappresenta il prezzo massimo unitario al quale risulta conveniente acquistare un'ulteriore risorsa

Dualità e analisi di sensitività (vi)

Un incremento pari a ϵ nel termine noto del primo vincolo corrisponde un incremento pari a 6.4ϵ nel valore ottimo della funzione obiettivo per qualunque valore di ϵ? **NO!**

Dualità e analisi di sensitività (vi)

Un incremento pari a ϵ nel termine noto del primo vincolo corrisponde un incremento pari a 6.4ϵ nel valore ottimo della funzione obiettivo per qualunque valore di ϵ? **NO!**

Informalmente, se ϵ è "grande" (in valore assoluto) è possibile che la soluzione ottima del problema duale cambi (non solo il valore, ma anche il punto ottimo, poiché la funzione obiettivo duale risulta modificata significativamente), e quindi che anche il prezzo ombra u_1^* sia differente.

Dualità e analisi di sensitività (vii)

esempio.mod

```
set PRODOTTI;
set RISORSE;

param prezzo{PRODOTTI} >= 0;
param consumo_risorse{RISORSE,PRODOTTI};
param disp{RISORSE} >= 0;

var x{PRODOTTI} >= 0;

maximize f : sum{i in PRODOTTI} prezzo[i]*x[i];

s.t. v_disp{i in RISORSE} :
  sum{j in PRODOTTI} consumo_risorse[i,j]*x[j] <= disp[i];
```

Dualità e analisi di sensitività (viii)

esempio.dat

```
set PRODOTTI := De_luxe Standard;
set RISORSE := I1 I2 Ore;

param prezzo :=
De_luxe 24
Standard 14;

param consumo_risorse : De_luxe Standard :=
I1 3 2
I2 4 1
Ore 2 1;

param disp :=
I1 1200
I2 1000
Ore 700;
```

esempio.run

```
reset;

model esempio.mod;
data esempio.dat;

option solver cplex;
solve;

display f;
display x;
```

Analisi si sensitività in AMPL

Per conoscere il valore delle variabili duali all'ottimo, in AMPL è sufficiente scrivere semplici istruzioni (senza quindi dover risolvere il problema duale).

Analisi si sensitività in AMPL

Per conoscere il valore delle variabili duali all'ottimo, in AMPL è sufficiente scrivere semplici istruzioni (senza quindi dover risolvere il problema duale).

Supponendo di aver definito un vincolo `constraint`, l'istruzione

```
display constraint;
```

permette di visualizzare il valore della variabile duale associata al vincolo.

Analisi si sensitività in AMPL

Per conoscere il valore delle variabili duali all'ottimo, in AMPL è sufficiente scrivere semplici istruzioni (senza quindi dover risolvere il problema duale).

Supponendo di aver definito un vincolo `constraint`, l'istruzione

```
display constraint;
```

permette di visualizzare il valore della variabile duale associata al vincolo.

Con CPLEX, attraverso il comando

```
option cplex_options 'sensitivity';
```

è possibile usare le istruzioni

`display constraint.down;` e `display constraint.up;`

per di visualizzare l'intervallo in cui è possibile variare il termine noto del vincolo `constraint` (lasciando gli altri inalterati) senza variare la base ottima.

Esercizi proposti (i)

Esercizio 4.1

- implementare in AMPL il problema primale e il problema duale dell'esempio
- verificare il valore ottimo delle variabili primali e duali
- verificare come cambia il valore ottimo della funzione obiettivo primale variando i termini noti dei vincoli
- verificare come cambia la soluzione ottima duale variando i termini noti dei vincoli primali
- visualizzare l'intervallo in cui è possibile variare il termine noto dei vincoli senza variare la base ottima

Esercizi proposti (ii)

Esercizio 4.2

L'azienda Free Time produce equipaggiamenti sportivi di due tipi: basic e avanzato. Ciascun equipaggiamento è costituito da una certa quantità di capi d'abbigliamento estivi, capi d'abbigliamento invernali e accessori. Tali quantità sono riportate nella seguente tabella.

	Basic	Avanzato
Numero capi invernali	2	3
Numero capi estivi	1	6
Numero accessori	4	5

Di seguito sono riportate le quantità disponibili dei capi d'abbigliamento e degli accessori.

Capi invernali	2200
Capi estivi	2000
Accessori	4200

I prezzi di vendita per i due equipaggiamenti sono di 10 e 50 Euro, rispettivamente. Determinare il piano produttivo che massimizza il ricavo. Inoltre, *supponendo le variabili continue*, determinare:

- se è vantaggioso per l'azienda aumentare (singolarmente) la quantità disponibile di capi invernali, capi estivi e accessori.
- il costo (unitario) massimo che l'azienda è disposta a pagare per un incremento (unitario) della quantità disponibile di capi invernali.
- l'incremento massimo di capi invernali che lascia invariata la base ottima del problema.

Esercizi proposti (iii)

sport.mod

```
set TIPI;
set COMPONENTI;

param Utilizzo{COMPONENTI,TIPI};
param Disp{COMPONENTI} >= 0;
param Prezzo{TIPI} >= 0;

var x{TIPI} >= 0 integer; #N.B.: l'analisi di sentitività
                          # va ristretta al caso continuo

maximize f : sum{j in TIPI} Prezzo[j]*x[j];

s.t. v_disp{i in COMPONENTI} :
sum{j in TIPI} Utilizzo[i,j]*x[j] <= Disp[i];
```

Esercizi proposti (iv)

sport.dat

```
set TIPI := Basic Avanzato;
set COMPONENTI := Estivi Invernali Accessori;

param Utilizzo : Basic Avanzato :=
Invernali   2      3
Estivi      1      6
Accessori   4      5;

param Disp :=
Invernali 2200
Estivi 2000
Accessori 4200;

param Prezzo :=
Basic 10
Avanzato 50;
```

Esercizi proposti (v)

sport.run

```
reset;
model sport.mod;
data sport.dat;

option solver cplex;
option relax_integrality 0; #considera variabili intere
option cplex_options '';
solve;
display f; display x;

option relax_integrality 1; #rilassamento continuo
                            # per analisi sensitività
option cplex_options 'sensitivity';
solve;
display f; display x;
display v_disp;
display v_disp.current, v_disp.down, v_disp.up;
```